

Song Liu, Mijun Hu, Jieli Fang, Cindy Song, Quting Zhang, BeiGene Corporation

## ABSTRACT

Dose intensity involves a couple of critical parameters: planned dose intensity (e.g. mg/m<sup>2</sup>/cycle in the sample study) which is planned dose per cycle as defined in the protocol; the actual dose intensity (ADI), defined as the total cumulative dose taken divided by the treatment duration. Without the Body Surface Area (BSA) being available in the electronic database (EDC), dose intensity derivation becomes complicated when we need to standardize the actual dose taken, EXDOSE (mg/m<sup>2</sup>/day) in the SDTM domain of EX. The treatment usually goes on continuously for several days and there could be dose interruptions in the middle. It becomes more challenging to calculate EXDOSE when we need to reset weight baseline after significant weight change happened. As a backup, we requested BSA at each visit from the blinded IRT data for reference when weight is missing. Since the chemotherapy includes different therapies, the treatment duration, total actual dose taken, total treatment cycle are all unique and critical to derive dose intensity. To save time and gain efficiency in table programming, we avoided using PARQUAL for each of the treatment within one ADEX, but created five ADEX<sub>med</sub>, where *med* represents each treatment following the same ADaM structure and PARAM/PARAMCD in each of the ADaM dataset.

## INTRODUCTION

Many oncology drugs are administered through infusion. For exposure analysis, we need to collect infusion/treatment start and end date/time, planned and actual dose taken, any change on planned dose, any AE that happened during drug administration, any dose delay and infusion interruption. This information is fundamental to derive actual dose taken in mg/m<sup>2</sup>/day, treatment duration, and actual dose intensity. We are using an example study to illustrate how to correctly map SDTM EX.EXDOSE and how to create ADaM ADEX for each of the treatment drugs.

This paper will discuss the derivation below and creation of ADaM ADEX<sub>med</sub>:

- Flag the first visit that weight change (increase or decrease) from the baseline is > 10%
- Reset weight baseline to calculate BSA correctly
- Derive treatment duration for 5FU treatment with the exact start/stop date and time
- Derive SDTM EX.EXDOSE (mg/m<sup>2</sup>/day) from EC.ECDOSE (mg)
- ADaM data specification for dose intensity related parameters
- Produce ADaM data for dose intensity related parameters

I hope this paper is straightforward for those with knowledge of the CDISC SDTM and ADaM standards to understand.

## DERIVING SDTM EX.EXDOSE FROM EC.ECDOSE

We are using chemotherapy 5-Fluorouracil (5-FU) as an example to show step by step on the calculation.

Table 1. 5-FU Treatment Information:

Intervention	Drug	Dose	Study Day(s)	Frequency of Administration	Route of Administration
Chemotherapy	5-Fluorouracil	750-800 mg/m <sup>2</sup>	1-5	Q3W	Intravenous

Table 2. SDTM EC:

USUBJID	ECTR	ECMOOD	ECOCCUR	ECDOSE	ECDOSU	VISIT	ECSTDTC	ECENDTC
xxx-001	5-FLUOROURACIL	SCHEDULED		750	mg/m2	CYCLE 1 DAY 1	2019-04-02	2019-04-07
xxx-001	5-FLUOROURACIL	PERFORMED	Y	6975	mg	CYCLE 1 DAY 1	2019-04-02T22:00	2019-04-07T21:42
xxx-001	5-FLUOROURACIL	SCHEDULED		750	mg/m2	CYCLE 2 DAY 1	2019-04-24	2019-04-29
xxx-001	5-FLUOROURACIL	PERFORMED	Y	6975	mg	CYCLE 2 DAY 1	2019-04-24T19:57	2019-04-29T20:44
xxx-001	5-FLUOROURACIL	SCHEDULED		750	mg/m2	CYCLE 3 DAY 1	2019-05-15	2019-05-20
xxx-001	5-FLUOROURACIL	PERFORMED	Y	6975	mg	CYCLE 3 DAY 1	2019-05-15T21:13	2019-05-20T18:42
xxx-001	5-FLUOROURACIL	SCHEDULED		750	mg/m2	CYCLE 4 DAY 1	2019-06-05	2019-06-10
xxx-001	5-FLUOROURACIL	PERFORMED	Y	6975	mg	CYCLE 4 DAY 1	2019-06-05T16:45	2019-06-10T12:00
xxx-001	5-FLUOROURACIL	SCHEDULED		750	mg/m2	CYCLE 5 DAY 1	2019-06-27	2019-07-02
xxx-001	5-FLUOROURACIL	PERFORMED	Y	6525	mg	CYCLE 5 DAY 1	2019-06-27T19:50	2019-07-02T19:30
xxx-001	5-FLUOROURACIL	SCHEDULED		750	mg/m2	CYCLE 6 DAY 1	2019-07-17	2019-07-22
xxx-001	5-FLUOROURACIL	PERFORMED	Y	6525	mg	CYCLE 6 DAY 1	2019-07-17T14:28	2019-07-22T14:30
xxx-001	5-FLUOROURACIL	SCHEDULED		750	mg/m2	CYCLE 7 DAY 1	2019-08-07	2019-08-12
xxx-001	5-FLUOROURACIL	PERFORMED	Y	6525	mg	CYCLE 7 DAY 1	2019-08-07T18:20	2019-08-12T17:37
xxx-001	5-FLUOROURACIL	SCHEDULED		750	mg/m2	CYCLE 8 DAY 1	2019-08-30	2019-09-04
xxx-001	5-FLUOROURACIL	PERFORMED	Y	6525	mg	CYCLE 8 DAY 1	2019-08-30T22:05	2019-09-04T18:32

The EC.ECDOSE (mg) is for the entire cycle where 5-FU was administered between Day 1 and 5 through continuous IV infusion over 24 hours. In order to convert ECDOSE to protocol specified daily dose in EX.EXDOSE (mg/m<sup>2</sup>/day), we need dividing total dose in mg by BSA and the treatment duration (day).

The protocol requires the following: the first doses of 5-FU are dependent upon the patient's baseline body weight. Subsequent doses must be recalculated if the change of body weight from baseline  $\geq 10\%$ . Subsequent doses should not be recalculated if the change (increase or decrease) of body weight from baseline  $< 10\%$  unless there is persistent toxicity that requires dose adjustment. If the dose is recalculated because of a  $\geq 10\%$  change in body weight from baseline, this body weight will then be used as the new baseline to calculate the platinum and 5-FU dose in subsequent cycles.

The data below shows that subject xxx-001 had decreased weight by 14.08% from baseline at Cycle 5 Day 1. The body weight of 61 kg at that visit became the new baseline weight, and the dose administered in subsequent visits had been adjusted from 6975 mg to 6525 mg based on the reduced body weight. When deriving BSA, reset baseline is the key to obtain the standardized dose in mg/m<sup>2</sup>/day.

Table 3. Weight change from baseline and BSA calculation:

USUBJID	WGT	WT_DATE	VISIT	HEIGHT	BASE_WT	Percent Change from Baseline	BSA	ABFL
xxx-001	71	2019-04-02	CYCLE 1 DAY 1	175	71	0.00	1.86	Y
xxx-001	67	2019-04-24	CYCLE 2 DAY 1	175	71	-5.63	1.86	
xxx-001	65	2019-05-15	CYCLE 3 DAY 1	175	71	-8.45	1.86	
xxx-001	65	2019-06-05	CYCLE 4 DAY 1	175	71	-8.45	1.86	
xxx-001	61	2019-06-27	CYCLE 5 DAY 1	175	61	-14.08	1.72	
xxx-001	60	2019-07-17	CYCLE 6 DAY 1	175	61	-1.64	1.72	
xxx-001	57	2019-08-07	CYCLE 7 DAY 1	175	61	-6.56	1.72	

xxx-001	59	2019-08-30	CYCLE 8 DAY 1	175	61	-3.28	1.72	
---------	----	------------	---------------	-----	----	-------	------	--

Weight can be missing or measured at different visits. We tried to use the BSA from the blinded IRT data. We noticed that BSA in IRT was not properly adjusted with the weight change and some of the BSA are missing. We decided to use the EDC data to derive BSA.

The code that detects the weight change, reset the weight for future treatment dose and convert SDTM EC.ECDOSE to EX.EXDOSE are below:

```

data bsa;
  set all;
  by usubjid wt_date;
  if first.usubjid then base_wt=.;
  retain base_wt;
  if abfl='Y' then base_wt=wgt;
  else if nmiss(base_wt,wgt)=0 then
    do;
      chg_b1=100*(wgt-base_wt)/base_wt;
      if abs(base_wt-wgt)/base_wt>=0.1 then base_wt=wgt;
    end;
  if nmiss(base_wt,height)=0 then bsa=sqrt(base_wt*height/3600);
run;

proc sort ;
  by usubjid visit;
run;

/** Convert ECDOSE to EXDOSE **/

data ec5fu;
  set sdtm.ec;
  where ectrt='5-FLUOROURACIL' and ec mood='PERFORMED';
  if ecstdtc ne '' then
    do;
      if find(ecstdtc,'T') then ecstdt=input(ecstdtc,e8601dt19.);
      else ecstdt=input(compress(ecstdtc||'T'||'00:00'),e8601dt19.);
    end;

  if ecendtc ne '' then
    do;
      if find(ecendtc,'T') then ecendt=input(ecendtc,e8601dt19.);
      else ecendt=input(compress(ecendtc||'T'||'00:00'),e8601dt19.);
    end;
  keep usubjid ecdose ecdosu visit ecstdt ecendt;
run;

proc sort ;
  by usubjid visit;
run;

data ex;
  merge ec5fu(in=a)
        bsa(keep=usubjid bsa visit);
  by usubjid visit;
  if a;
  txdur=(ecendt-ecstdt)/(60*60*24);
  exdose=ecdose/txdur/bsa;
  if not missing(exdose) then exdosu='mg/m2/day';
  cycle=input(scan(visit,2,' '),best.);

  format ecendt ecstdt datetime19. txdur exdose bsa 8.2;
run;

```

Table 4. EC.ECDOSE is converted to EX.EXDOSE

USUBJID	ECDOSE	ECDOSU	VISIT	EXSTDTC	EXENDTC	BSA	TXDUR	EXDOSE	EXDOEU
xxx-001	6975	mg	CYCLE 1 DAY 1	02Apr2019 22:00:00	07Apr2019 21:42:00	1.86	4.99	752.77	mg/m2/day
xxx-001	6975	mg	CYCLE 2 DAY 1	24Apr2019 19:57:00	29Apr2019 20:44:00	1.86	5.03	746.02	mg/m2/day
xxx-001	6975	mg	CYCLE 3 DAY 1	15May2019 21:13:00	20May2019 18:42:00	1.86	4.90	766.98	mg/m2/day
xxx-001	6975	mg	CYCLE 4 DAY 1	05Jun2019 16:45:00	10Jun2019 12:00:00	1.86	4.80	781.84	mg/m2/day
xxx-001	6525	mg	CYCLE 5 DAY 1	27Jun2019 19:50:00	02Jul2019 19:30:00	1.72	4.99	759.95	mg/m2/day
xxx-001	6525	mg	CYCLE 6 DAY 1	17Jul2019 14:28:00	22Jul2019 14:30:00	1.72	5.00	757.63	mg/m2/day
xxx-001	6525	mg	CYCLE 7 DAY 1	07Aug2019 18:20:00	12Aug2019 17:37:00	1.72	4.97	762.39	mg/m2/day
xxx-001	6525	mg	CYCLE 8 DAY 1	30Aug2019 22:05:00	04Sep2019 18:32:00	1.72	4.85	780.94	mg/m2/day

## CREATING ADAM DATA FOR EXPOSURE ANALYSIS

The study involves five different chemotherapies that each treatment collected similar information about drug administration. For exposure analysis, we created a set of PARAM/PARAMCD with the same names in each of the exposure analysis datasets, and named them *ADEXmedA*, *ADEXmedB*, *ADEXmedC*, *AEXmedD* and *ADEXFLU*.

Table 5. 5-FU Dose Intensity Related Specification in ADEXFLU:

PARAMCD	PARAM	DERIVATION
<b>ACUMDOSE</b>	ACTUAL CUMULATIVE DOSE (MG/M2)	Sum of all dose received at each cycle in mg/m2. Total actual cumulative doses will be sum of EX.EXDOSE (mg/m2/day) multiply treatment duration, which is (input(EXENDTC,e8601dt.) - input(EXSTDTC,e8601dt.)) / (24*3600) at each cycle with EX.EXTRT="5-FLUOROURACIL" when treatment start and end date time not missing. Otherwise, treatment duration is (input(EXENDTC,e8601da.) - input(EXSTDTC,e8601da.)) when treatment time is missing.
<b>LDOSEN</b>	CYCLE NUMBER OF LAST DOSING	This is the number of cycles on last dosing CRF page. Set to the last treatment cycle when EC.ECMOOD="PERFORMED" and EC.ECDOSE > 0 and EC.ECTRT="5-FLUOROURACIL"
<b>ATDOSINT</b>	ACTUAL DOSE INTENSITY (MG/M2/CYCLE)	Cumulative dose/Treatment Duration (cycle). Set to ACUMDOSE/treatment duration, where treatment duration is $\max((\text{last dose date} + (21-5) - \text{first dose date} + 1) / 21, \text{LDOSEN})$ , specific to 5-FU treatment.
<b>RLDOSINT</b>	RELATIVE DOSE INTENSITY (%)	Actual Dose Intensity/Planned Dose Intensity*100

Actual Dose Intensity (ADI) could also be calculated in mg per dose, mg/week, or other appropriate units. In this sample, we are using mg/m<sup>2</sup>/cycle. It is important to get the treatment duration (TD) correct:

- Since 5-FU is given on days 1-5 of each 21-day cycle, treatment duration is the last dose date +(21-5) – first date of treatment + 1. We are adding cycle length and subtract the five-day dosing period.
- Divide treatment duration by the cycle length 21 to get the derived number of treatment cycles.
- Subject may not complete the five-day dosing in the last dosing cycle, using the last treatment date may get the total treated cycle short, so we are taking the maximum between the derived cycle from treatment date and the total number of treated cycles observed as the denominator.

The relative dose intensity (RDI) involves two important components:

1. Actual Dose Intensity (ADI)
2. Planned Dose Intensity (PDI)

Relative Dose Intensity (RDI) = Actual Dose Intensity/Planned Dose Intensity\*100 (%)

Planned dose is the one as marked on eCRF at Cycle1 Day 1, which is 750 \* 5 mg/cycle in this example.

Table 6. Relative Dose Intensity (RDI) algorithm:

	ADI (mg/m <sup>2</sup> /cycle)	Planned Dose per Cycle	RDI
5-FU	$\frac{\sum_1^{\# \text{ of cycles}} \frac{\text{actual dose}}{BSA *}}{\max\left(\frac{\text{date of last dose up to cutoff} + 17 - \text{first dose date}}{21}, \text{total number of treated cycles}\right)}$	750 * 5 mg/m <sup>2</sup>	$\frac{ADI}{3750}$

ADaM program deriving the above parameters:

```

/*****ACUMDOSE: ACTUAL CUMULATIVE DOSE (MG/M2)*****/
data ACUMDOSE;
length PARAMCD $8 PARAM $200;
set ex(where=(exdose ne .));
by usubjid visit;
if first.usubjid then AVAL=exdose*txdur;
    else AVAL+exdose*txdur;
if last.USUBJID;
PARAMCD='ACUMDOSE';
PARAMN=1;
PARAM='ACTUAL CUMULATIVE DOSE (MG/M2)';
keep USUBJID PARAMCD AVAL PARAMN PARAM;
run;

/*****LDOSEN:CYCLE NUMBER OF LAST DOSING*****/

proc sql noprint;
create table LDOSEN as
select distinct usubjid,input(scan(visit,2,' '),best.) as aval,
2 as PARAMN,'LDOSEN' as PARAMCD length=8, 'NUMBER OF LAST DOSING' as
PARAM length=200
from ex
where exdose>0
group by usubjid
having cycle=max(cycle);
quit;

/*****ATDOSINT: ACTUAL DOSE INTENSITY (MG/M2/CYCLE)*****/
data ATDOSINT;
merge ACUMDOSE(in=a rename=(AVAL=acumdose))
ADAM.ADSL(in=b keep=USUBJID FLUEDT FLUSDT)
/****FLUSDT:date of first dose of 5-FU, FLUEDT:date of last dose of 5-FU *****/
LDOSEN(rename=(AVAL=cycle));
by USUBJID;
if a & b ;
length PARAMCD $8 PARAM $200;
/****if dosing is not complete in the last dosing cycle, the actual dose intensity
should be considered as actual dose divided by full cycle*****/
full_cyc=floor(max((FLUEDT+17-FLUSDT)/21,cycle));
if nmiss(acumdose,full_cyc)=0 then aval=acumdose /full_cyc;

```

```

PARAMCD='ATDOSINT';
PARAMN=3;
PARAM='ACTUAL DOSE INTENSITY (MG/M2/CYCLE)';
keep USUBJID PARAMCD AVAL PARAMN PARAM;
run;

/*****RLDOSINT : RELATIVE DOSE INTENSITY (%)*****/
data RLDOSINT;
set ATDOSINT(drop=PARAMCD rename=(AVAL=atdosint));
AVAL=100*atdosint/3750;
length PARAMCD $8 PARAM $200;
PARAMCD='RLDOSINT';
PARAMN=4;
PARAM='RELATIVE DOSE INTENSITY (%)';
keep USUBJID AVAL PARAMCD PARAMN PARAM;
run;

```

Table 7. 5-FU Dose Intensity Related parameters in ADEXFLU:

USUBJID	PARAMN	PARAMCD	PARAM	AVAL
xxx-001	1	ACUMDOS	ACTUAL CUMULATIVE DOSE (MG/M2)	30174.64
xxx-001	2	LDOSEN	CYCLE NUMBER OF LAST DOSING	8
xxx-001	3	ATDOSIN	ACTUAL DOSE INTENSITY (MG/M2/CYCLE)	3771.83
xxx-001	4	RLDOSIN	RELATIVE DOSE INTENSITY (%)	100.58

## CONCLUSION

Dose intensity is an important part of exposure analysis. Weight change directly impacts the derived EX.EXDOSE (mg/m<sup>2</sup>/day), so to determine the visit where the weight change ≥10% and reset the weight baseline is critical for the correct BSA and EXDOSE calculation. Using the actual treatment start and end time to derive treatment duration in days is more accurate than using the date interval at the visit level. The key for actual dose intensity (ADI) is the treatment cycle of the specific drug. Different therapy can have different length of treatment cycle. If the treatment is just one day with a cycle of 21, you want to add 21 days from the last dose date then minus the first dose date to get the treatment cycle for ADI calculation. Since we have five treatments in the sample study, we put each dosing data in separate ADaMs while keeping identical PARAM/PARAMCD names in the ADaM. This has ultimately increased efficiency on ADaM and TFL production for exposure analysis.

## REFERENCES

CDISC Submission Data Standards Team. (2013). Study Data Tabulation Model Implementation Guide: Human Clinical Trials Version 3.2. CDISC.

CDISC ADaM Team. 2019. "ADaM Implementation Guide v1.2". Accessed April 20, 2019.

<https://www.cdisc.org/standards/foundational/adam>

## ACKNOWLEDGMENTS

Thank Zhongan (Stephen) Chen to be the technical reviewer and his valuable input.

## CONTACT INFORMATION

Your comments and questions are valued and encouraged. Contact the author at:

Song Liu  
BeiGene Inc  
[Song.liu@beigene.com](mailto:Song.liu@beigene.com)

SAS and all other SAS Institute Inc. product or service names are registered trademarks or trademarks of SAS® Institute Inc. in the USA and other countries. ® indicates USA registration.

Other brand and product names are trademarks of their respective companies